

## Session 7: Multimedia & GPS

At the end of this session participants will be able to:

- Launch external applications from a data entry application
- Launch photo and document viewers from a data entry application
- Capture GPS coordinates in a data entry application
- Display captured GPS points on a map
- Generate and view reports from a data entry application

### Launching External Applications on Windows

The command `execsystem()` lets you launch another application from logic. It takes the name of the application to launch. For example to add a userbar button to bring up the calculator on Windows you could do:

```
PROC GLOBAL

// Launch the calculator
function showCalculator()
    execsystem("calc.exe");
end;

PROC SURVEY_FF
preproc
userbar(clear);
userbar(add button, "Calculator", showCalculator);
userbar(show);
```

For Windows utilities like calculator and notepad which are in the system directory you can simply give the name. For other applications that are installed in the Program Files directory you may need to specify the full path. For example to launch Microsoft Word (Office 2010 version) you would use:

```
execsystem("C:\Program Files\Microsoft Office\Office14");
```

### Launching External Applications on Android

The `execsystem()` command is slightly different on Android. On Android to launch an application with `execsystem()` use “app:” followed by the package name of the application. For example to launch Gmail:

```
execsystem("app:com.google.android.gm");
```

Finding the package name can be a bit tricky. The easiest way is to use your web browser to search for the application on the Google Play website ([play.google.com](http://play.google.com)). On the page for the application the

package name will be the last part of the address in the address bar of your browser, after the “?id=”. For example if you go the page for CSEntry on the Google Play Store the address is:

*<https://play.google.com/store/apps/details?id=gov.census.cspro.csentry>*

The package name is therefore “gov.census.cspro.csentry”.

Going back to launching the calculator from Android, unfortunately different Android devices have different calculators so you have try multiple package names. The following code tries the two different versions that we have seen:

```
// Samsung uses app:com.sec.android.app.popupcalculator,  
// LG and HTC use app:com.android.calculator2.  
// We will try one and if it doesn't work then try the other.  
if execsystem("app:com.sec.android.app.popupcalculator") <> 1 then  
    if execsystem("app:com.android.calculator2") <> 1 then  
        errmsg("Unable to launch calculator.");  
    endif;  
endif;
```

To make our code work on both Android and Windows we can use the function *getos()* which returns a number between 10 and 19 for Windows and between 20 and 29 for Android.

```
// Launch the calculator  
function showCalculator()  
    if getos() in 10:19 then  
        // Windows  
        execsystem("calc.exe");  
    else  
        // Android - different devices have different built in calculators.  
        // Samsung uses app:com.sec.android.app.popupcalculator,  
        // LG and HTC use app:com.android.calculator2.  
        // We will try one and if it doesn't work then try the other.  
        if execsystem("app:com.sec.android.app.popupcalculator") <> 1 then  
            if execsystem("app:com.android.calculator2") <> 1 then  
                errmsg("Unable to launch calculator.");  
            endif;  
        endif;  
    endif;  
end;
```

## Viewing Files on Android

On Android *execsystem()* can also be used to launch files with their default viewer.

```
execsystem("view:/mnt/sdcard/csentry/picture.jpg");  
execsystem("view:/mnt/sdcard/csentry/audio.mp3");  
execsystem("view:/mnt/sdcard/csentry/movie.mp4");  
execsystem("view:/mnt/sdcard/csentry/document.pdf");
```

For this to work, the device must have an application installed capable of viewing the type of file specified. Android determines which application to launch to view the file based on the file extension. All Android devices have viewers for most photo, music and video file formats, however not all have a built-in PDF viewer. If you do not have one on your device you can always download one from Google Play.

Let's add a usebar button to show a photo of the different kinds of sweet potatoes as a visual aid for question G4:

```
// Display a picture of a sweet potato
function showSweetPotato()
    execsystem("view:/mnt/sdcard/csentry/Survey/photos/sweetpotato.jpg");
end;
```

When we test this on Android, make sure to copy the photo onto the Android device and make sure it is in the photos subdirectory of the Survey subdirectory of the CSEntry directory (Survey/photos/sweetpotato.jpg).

We can make our code a little more flexible by using the function *pathname()* to get the directory where the application is stored. This way if someone copies it into a different folder on the phone/tablet it will still work:

```
// Display a picture of a sweet potato
function showSweetPotato()
    execsystem(makertext("view:%sphotos/sweetpotato.jpg",
                        pathname(Application)));
end;
```

## Taking Photos on Android

To take a picture on Android you can use *execsystem* with "camera:" followed by the full path in which to save the photo:

```
execsystem("camera:/mnt/sdcard/csentry/photo.jpg");
```

Note that photos do not get saved into the application data file. You need to save them to a folder/file on the device that will make it easy to link the photo back to the interview that it was taken for. The easiest way to do this is to make the name of the photo based on the id-items of the questionnaire:

```
string photoFilename = makertext("photo%d%02d%03d.jpg",
                                DISTRICT, VILLAGE, HOUSEHOLD_NUMBER);
execsystem(makertext("camera:/mnt/sdcard/csentry/%s", photoFilename));
```

Let's have the interviewer take a photo of the household being interviewed in section A. Rather than adding a userbar button which would be easy for the interviewer to forget, we will make it part of the questionnaire by adding a dummy variable to the dictionary. In the postproc of the variable we will call

`execsystem()` to take the photo. We can add a refused code in case the household doesn't want a picture taken.

```
PROC PHOTO

if PHOTO = 1 then
    // Take photo
    string photoFilename = maketext("%sphoto%d%02d%03d.jpg",
                                    pathname(Application),
                                    DISTRICT, VILLAGE, HOUSEHOLD_NUMBER);
    execsystem(maketext("camera:%s", photoFilename));
elseif PHOTO = 9 then
    // refused, move to next field with no photo
endif;
```

This works the first time through the questionnaire but if we go back through the photo question a second time then we are forced to take another photo. We can avoid this by adding another option to the value set “Keep photo” that the interviewer will select once they are happy with the photo that they took. When “keep photo” is selected we won't launch the camera. While we are it we can add another option “View photo” to let them look at the photo that was taken.

```
PROC PHOTO

string photoFilename = maketext("%sphotos/photo%d%02d%03d.jpg",
                                pathname(Application),
                                DISTRICT, VILLAGE, HOUSEHOLD_NUMBER);

if PHOTO = 1 then
    // Take/retake photo
    execsystem(maketext("camera:%s", photoFilename));

    // reenter so that interview can retake if it is not good.
    reenter;
elseif PHOTO = 2 then
    // view existing photo
    execsystem(maketext("view:%s", photoFilename));

    // reenter so that interview can retake if it is not good.
    reenter;
elseif PHOTO = 3 then
    // Keep photo - move to next field
elseif PHOTO = 9 then
    // No photo (refused)
    // Delete photo if it exists
    filedelete(photoFilename);
endif;
```

This works except that it is possible for the interviewer to select “Keep photo” even when there is no existing photo and to attempt to view a photo that doesn't exist. We can solve this by only showing the “Keep photo” and “View photo” options in the value set if the photo exists already.

```

PROC PHOTO
onfocus
string photoFilename = maketext("%sphotos/photo%d%02d%03d.jpg",
                                pathname(Application),
                                DISTRICT, VILLAGE, HOUSEHOLD_NUMBER);

// Only show the view and keep options in value set
// if the photo exists.
if fileexist(photoFilename) then
    setvalueset(PHOTO, PHOTO_VS_PHOTO_TAKEN);
else
    setvalueset(PHOTO, PHOTO_VS_NO_PHOTO);
endif;

```

## Viewing GPS Points

On Android you can use *execsystem* to display a GPS point on a map using “gps:” followed by the latitude and longitude. We can add a userbar button to display the household location on a map:

```

// Display household GPS point on map
function showHouseholdOnMap()
    execsystem(maketext("gps:%f,%f", LATITUDE, LONGITUDE));
end;

```

## Capturing GPS Points

To capture a GPS point first you need to start up the GPS hardware:

```
gps(open);
```

Then you request a GPS reading giving it a timeout in seconds and optionally a desired accuracy in meters:

```

if gps(read, 60, 10) then // Read up to 60 seconds, try for 10m accuracy
    errmsg("Latitude is %f, longitude is %f", gps(latitude), gps(longitude));
else
    errmsg("GPS signal could not be acquired");
endif;

```

Finally you close the GPS:

```
gps(close);
```

In addition to querying the GPS latitude and longitude you can also query the accuracy, number of satellites, and altitude (although on Android altitude is not very accurate). See the help on GPS for details.

The longitude and latitude reported are in degrees in the WGS84 datum and are returned as decimal numbers.

Let's add a user button to update the GPS coordinates for the household:

```
// Capture current household location using GPS
function getGPS()
  gps(open);
  if gps(read, 60) then // Read for up to 60 seconds
    LATITUDE = gps(latitude);
    LONGITUDE = gps(longitude);
  else
    errmsg("GPS signal could not be acquired");
  endif;
  gps(close);
end;
```

If we want to be fancy, we can display the point on the map first so that the interviewer can view it and then decide whether or not to use it.

```
// Capture current household location using GPS
function getGPS()
  gps(open);
  if gps(read, 60) then // Read for up to 60 seconds
    // Show map so that interview can see result.
    execsystem(maketext("gps:%f,%f",
      gps(latitude), gps(longitude)));
    if accept("Save this result", "Yes", "No") = 1 then
      LATITUDE = gps(latitude);
      LONGITUDE = gps(longitude);
    endif;
  else
    errmsg("GPS signal could not be acquired");
  endif;
  gps(close);
end;
```

## Group Exercise: Plots

Form teams of 3-4 for this exercise. Implement section I: Plots in the questionnaire.

- Question I1: use a dynamic value set based on the crops produced from question G1.
- Questions I2 and I3: take a GPS reading using the *gps()* command and record the latitude and longitude automatically.
- Question I4: have the interviewer take a photo and store it in the photos directory with a unique name that includes the id-items and the plot number, for example plot1-02-003-4.jpg for district 1, village 2, household 3, plot 4.
- Question I5: use a separate area measurement calculator application that you will download from Google Play. A couple of options are *Distance and area measurement* by Kristofer Björnson and *Area Calculator Free* by Antti Pouru. You should launch this application using *execsystem*

when question I5 is entered. The interviewer will use it to measure the area and then return to CSEntry and manually type the area into question I5.

## Writing and Viewing Reports

With `execsystem()` on Android you can view web pages using “browse:” followed by the address:

```
execsystem("browse:https://www.census.gov");
```

We can use this to view web pages (HTML documents) and text documents that we write out from CSEntry. To write out files from a CSEntry application we have to first declare a file variable in the PROC GLOBAL:

```
PROC GLOBAL  
file tempFile;
```

Then we can use the commands `setfile()`, `filewrite()` and `fileclose()` to open, write to and close the file. Let's create a userbar button to write a simple text report that lists the household members:

```
// Write out and display household summary report  
function showHouseholdReport()  
    string reportFilename = maketext("%sreport.txt", pathname(Application));  
    setfile(tempFile, reportFilename);  
    filewrite(tempFile, "Household Summary Report");  
    filewrite(tempFile, "-----");  
    filewrite(tempFile, "");  
    filewrite(tempFile, "District %d Village %d Household Number %d",  
              visualvalue(DISTRICT),  
              visualvalue(VILLAGE),  
              visualvalue(HOUSEHOLD_NUMBER));  
    filewrite(tempFile, "");  
    filewrite(tempFile, "Members 5 years and over:");  
    filewrite(tempFile, "");  
    numeric i;  
    filewrite(tempFile,  
    "Name           Sex      Age  Relationship");  
    filewrite(tempFile, "----          ---      ---  -----  
---");  
    do i = 1 while i < totocc(ADULTS000)  
        filewrite(tempFile, "%s %-6s %3d %s",  
                  NAME(i),  
                  getlabel(SEX, visualvalue(SEX(i))),  
                  visualvalue(AGE(i)),  
                  getlabel(RELATIONSHIP_VS1,  
visualvalue(RELATIONSHIP(i))));  
    enddo;  
    close(tempFile);  
    if getos() in 20:29 then  
        // Android - use "view:"  
        execsystem(maketext("view:%s", reportFilename));  
    else
```

```

        // Windows - use "explorer.exe <filename>"
        execsystem(makertext("explorer.exe %s", reportFilename));
    endif;
end;

```

On Windows we can't use *execsystem* with "view:" so instead we launch Windows Explorer (explorer.exe) with the name of the file and that will open the file in notepad for a text file and in the web browser for an HTML file.

If you are familiar with HTML you can write out HTML and have nicer formatting and pictures:

```

// Write out and display household summary report using HTML
function showHouseholdReportHTML()
    string reportFilename = makertext("%sreport.html", pathname(Application));
    setfile(tempFile, reportFilename);
    fwrite(tempFile, "<!DOCTYPE html>");
    fwrite(tempFile, "<html><head>");
    fwrite(tempFile, "<style type='text/css'>");
    fwrite(tempFile, "table, th, td {border: 1px solid black;border-
collapse: collapse;padding: 8px}");
    fwrite(tempFile, "</style>");
    fwrite(tempFile, "<title>Household Summary Report</title>");
    fwrite(tempFile, "</head>");
    fwrite(tempFile, "<body>");

    fwrite(tempFile, "<h2>Household Summary Report</h2>");
    fwrite(tempFile, "<p>District: %d</p>",
        visualvalue(DISTRICT));
    fwrite(tempFile, "<p>Village: %d</p>",
        visualvalue(VILLAGE));
    fwrite(tempFile, "<p>Household Number %d</p>",
        visualvalue(HOUSEHOLD_NUMBER));
    fwrite(tempFile, '',
        visualvalue(DISTRICT), visualvalue(VILLAGE),
        visualvalue(HOUSEHOLD_NUMBER));
    fwrite(tempFile, "<p>Members 5 years and over:</p>");
    numeric i;
    fwrite(tempFile, "<table>");
    fwrite(tempFile,
"<tr><th>Name</th><th>Sex</th><th>Age</th><th>Relationship</th></tr>");
    do i = 1 while i < totocc(ADULTS000)
        fwrite(tempFile, "<tr>");
        fwrite(tempFile, "<td>%s</td>", NAME(i));
        fwrite(tempFile, "<td>%s</td>",
            getlabel(SEX, visualvalue(SEX(i))));
        fwrite(tempFile, "<td>%d</td>", visualvalue(AGE(i)));
        fwrite(tempFile, "<td>%s</td>",
            getlabel(RELATIONSHIP_VS1,

```



```

        visualvalue(RELATIONSHIP(i))));
    fwrite(tempFile, "</tr>");
enddo;
fwrite(tempFile, "</table>");
fwrite(tempFile, "</body></html>");
close(tempFile);
if getos() in 20:29 then
    // Android - use "view:"
    execsystem(maketext("view:%s", reportFilename));
else
    // Windows - use "explorer.exe <filename>"
    execsystem(maketext("%sexplorer.exe %s",
        pathname(Windows),
        reportFilename));
endif;
end;

```

## Exercises

1. Add a button to the userbar that opens your website (the website for your organization). Test this on Android. Bonus if you can get it to work on Windows too.
2. Add a button to the userbar that shows the interviewer manual. You can use the PDF interviewer manual that is in the Dropbox (it only has a cover page but is enough for this exercise).
3. Add the following additional information to the summary report. Add it to the text version unless you are already an HTML expert in which case add it to the HTML version:
  - a. Total number of household members 5 and over, total males 5 and over, total females 5 and over.
  - b. List of assets with quantity and owner. Do not list assets where the quantity is zero.